



Powered by

ENVIANCE

© 2017, Enviance Inc. rev 7/7/2017

10 Example KeyControl Hotkeys Using RSIScript

Ergonomics often focuses on improving posture, workstation positioning, and other environmental factors. But the other half of the ergonomics equation is *reducing how much you need to do*. RSI Guard has tools like KeyControl and AutoClick to help you reduce mouse and keyboard activity while being equally or more productive.

For example, KeyControl lets you to assign many standard functions (e.g. open a web page, double-click, type standard text) to a hotkey. This make performing these functions faster and easier – and that’s good both for health and productivity.

If you want to get even more out of KeyControl, you can also make hotkeys that perform complex and/or multi-step processes. This can be done using RSI Guard’s built in scripting language, RSIScript. This document provides 10 examples of common activities you might want to assign to a hotkey that require using RSIScript.

These examples are just a taste! RSIScript is very flexible, and almost any repetitive process you do can be automated. If you want to dive deeper into RSIScript, visit <http://www.rsiguard.com/rsiscript>

The Examples

1) Type text, including foreign characters and special characters.....	3
2) Send a standard email with a custom Subject line.....	3
3) Complete a form using logic/context/user-input.....	4
4) Open multiple websites.....	4
5) Open an application instance or activate an already open one	5
6) Create a quick-launcher for multiple applications/websites/files.....	5
7) Open the OES Dashboard	5
8) Use Text-to-speech to speak the selected text.....	5
9) Automate a series of mouse and typing events.....	6
10) Find a button on the screen and move the mouse to it.....	6

How to use the examples in this document

The following steps explain how to implement the examples in this document.

- 1) The examples below address specific tasks. You'll need to customize most of them to apply to your particular situation. So, first, figure out which example is closest to what you want to do, and start there.
- 2) Copy the scripts and paste them into a text editor (e.g. Notepad for Windows, TextEdit for Mac). Note that this document wraps text, but where text is wrapped (hanging indent), your script must actually contain the entire command on one single line.
- 3) Edit the scripts as needed to meet your goals for your hotkey.
- 4) Save your script. You may want to save it to a network or Cloud drive if you want to use your hotkey on multiple computers. Otherwise, you could save it in a subfolder of your local Documents folder.
- 5) In RSIGuard, click Setup, Settings, and select the KeyControl tab.
- 6) Click the "Add a New Hotkey" button and press the key-combination you want to use to trigger your new script.
- 7) Under "Select A Category", select "Run RSIScript".
- 8) Click Browse, and browse to the text file (.TXT) you saved.
- 9) Give your hotkey a descriptive name.
- 10) Click OK to complete hotkey, then OK in settings screen to save & activate your hotkey.

Visit KeyControl help at <http://www.rsiguard.com/help/helpkc.htm> for more information.

1) Type text, including foreign characters and special characters

If you type certain text repetitively (e.g. an address, a phrase you commonly use in emails, etc.) then you may want to assign that text to a hotkey. Although the basic “Add a Hotkey” feature allows you to do this with static text, using RSIScript allows you to do more complex typing. Here is an example:

```
# type when the request was received (use current date)
type Sit-stand desk request was received on $DateStr(%A %B %d).<enter>
# turnaround time is 2 weeks, so type that too
type You should receive your desk by $DateStr(%A %B %d,$Date(14)).<enter>
```


This would type something like the following into your active email/document:

```
Sit-stand desk request was received on Monday May 08.
You should receive your desk by Monday May 22.
```

To type special characters, see: <http://rsiguard.com/help/helpkcchars.htm>

2) Send a standard email with a custom Subject line

In this example, we imagine you are tabbed to a field where a user’s email address is shown:



The screenshot shows a form with two input fields. The top field is labeled "Request Date:" and contains the date "2017/05/01". The bottom field is labeled "Sit-stand desk request from:" and contains the email address "jjaneuser@acme.com".

We want to send an email to the user letting them know their request was approved or denied, and inform facilities as well.

```
# select the email address via Ctrl+A
type <ctrla>
# copy the email address via Ctrl+C
type <ctrlc>
# start building a new email - begin by specifying CC address (to facilities)
mail cc facilities@acme.com
# specify which of 3 subjects to use by querying user
mail subject Your request was $choice(Subject,approved,denied)
# initiate the new email to the email address in copy buffer
mail sendto $clipboardtext()
#wait a second for email to open
wait 1
#new email starts with cursor in Message area, so start typing
type Thanks for your request. If you need additional information...
type Sincerely, Staff
```

3) Complete a form using logic/context/user-input

You can complete a form simply using a typing hotkey. E.g., say you need to type 'first name' and 'last name' in adjacent fields, you could simply use a typing hotkey with the text:

```
Jane<tab>Smith<tab>
```

This would type 'Jane', Tab to the last name field, type 'Smith', and then Tab to move the next field.

But let's say you want to enter different values based on user input and/or the values of fields or dates. This example RSIScript completes a more complex form. Read the comments to follow what it does:

```
#go to requisition form
open http://www.rsiguard.com/pp/misc/requisition.htm
#tab to date field (how you do this can vary by browser)
type <tab>
#type request date (today)
type $DateStr(%Y-%m-%d)<tab>
#type requestor's ID (username)
type $u<tab>

#ask user what is being requested
setvar 1 $choice(Specify requested item,Sit-Stand Desk,Monitor Arm,Petite
  Chair,Large Chair,Adjustable Keyboard,Small Mouse)
#type first 2 characters (enough to uniquely identify the entry in the drop-
  down list)
type $Left($1,2)<tab>

#current field tells calculated turnaround time for requested item - copy it
type <ctrlc>
#prepare message for submitting request, warn if turnaround time is over 5 days
setvar 2 $if($greaterthan($clipboardtext(),5),Request for $1 has turnaround
  time of $clipboardtext() days.) Click this notification to cancel request.
#have user click notification to complete request (activate form, submit it,
  wait 1s, close tab)
notice msg FF0000 10 "$1 Request" "Click this notification to complete the
  request" {{$do(activate Requisition,type <tab><enter>,wait 1,type
  <ctrl<f4>>)}}
#if user doesn't click notification, they can manually submit it, or just exit
  the requisition page
```

4) Open multiple websites

You can easily make a hotkey open a single website with KeyControl. But you can also make a hotkey open a set of websites. It can use logic (e.g. day of the week) to open different pages.

```
#open 2 pages I always need
open http://www.rsiguard.com
open http://www.enviance.com

#only open this last page on Fridays
$if($equal($DateStr(%A),Friday),open http://www.rsiguard.com/pp/misc/friday.htm)
```

Internet Explorer may sometimes open both websites in the same tab. If you only see the last window, you can use the following:

```
open http://www.rsiguard.com
schedule in 1 open http://www.enviance.com
schedule in 2 open http://www.google.com
```

5) Open an application instance or activate an already open one

With “Add a New Hotkey” you can make a hotkey to open an application, e.g. Excel. But a more sophisticated hotkey will open Excel if it’s not already running, or activate an already-running instance of Excel.

```
#get the number of Excel windows that are running
setvar 1 $isrunning(Excel)
#if it's zero, open a new instance
$if($equal($1,0),open Excel)
#otherwise, activate an Excel window
activate Excel
```

6) Create a quick-launcher for multiple applications/websites/files.

Let’s say you have 10 places you visit often. You could create 10 hotkeys. But that might be hard to remember. An alternative would be to create one hotkey that gives you quick access to the 10 places. Here are 2 strategies:

- 1) You could create a hotkey that opens a folder that contains Shortcuts. This option is easy to manage. If you want to launch something, just press the hotkey, and the folder opens. Use the appropriate Shortcut to launch the application/website/file. You can easily add/remove shortcuts (e.g. by Right clicking, selecting New, Shortcut or Right-button drag/dropping the item you want into the folder). The RSIScript is simply ‘open FOLDER’ where FOLDER is the path to your shortcuts folder. For example, if you put it created the folder C:\MyShortcuts, you’d use ‘open C:\MyShortcuts’.
- 2) You could use the \$choice() and/or \$switch() commands to present the user a popup with choices. This method is a little more complex to set up and manage, but it lets you access all the options with 1 keypress (in addition to pressing the hotkey).

7) Open the OES Dashboard

This hotkey uses a web API call to get the URL for the dashboard, and then opens the dashboard.

```
#Bring RSIGuard to foreground (use the extra - to help insure you get the
RSIGuard application)
activate RSIGuard-
#use the response from an API to get the URL for your dashboard
setvar 1 $api(http://www.rsiguard.com/pp/misc/opensdashboard.txt)
#open the dashboard in a 900X900 Window without a separate Close Button
html open "My Dashboard" "" 900 900 $1
```

8) Use Text-to-speech to speak the selected text

RSIGuard can speak specified text using this RSIScript command:

```
speak text This is a sample of text to speech functionality.
```

Alternately, you can use the following command to speak all the text you’ve selected.

```
speak selected
```

This lets you select text in a document or on a web page and have the computer read it aloud to you.

9) Automate a series of mouse and typing events

Although you can create hotkeys in “Add a New Hotkey” to type text or do mouse clicks, sometimes you need to automate a series of mouse and typing events. You can do this using the ‘mouse’ and ‘type’ commands.

For example, this script activates a known application, moves the mouse to a specific location in the application, clicks, and then types text (presumably into some field).

```
#activate the application of choice
activate ACMEApp

#move to a specific location within the application
mouse absmove client 400 300

#click the mouse (e.g. to activate a field)
mouse click


#type some text in that field
type Hello
```

10) Find a button on the screen and move the mouse to it

Often the processes we want to automate require locating a button or image on the screen that isn’t in a fixed location. Or, figuring out what to do next is dependent on the presence of some visual indicator.

RSIGuard’s scripting language contains a basic visual recognition tool to find graphical objects on the screen. This could be used in the previous example if the field you wanted to type into wasn’t necessarily always at the same coordinates.

This is complex functionality, and this documentation only provides a basic example. Creating hotkeys to automate tasks using this functionality require Enviance support or an experienced software engineer.

The most basic usage of this functionality starts with creating a BMP image file of a very small part of the object you want to find. For example, say you want to click on the Send button in an email application. You might create a BMP of a portion of the Send button like:  (The image must match in size exactly.)

Then, in RSIScript, you use an expression like:

```
setvar 1 $finding(c:\MyImages\SendButton.bmp,curwnd)
```

The variable \$1 will now either equal 0 if the image wasn’t found, or coordinates (e.g. 400 300) if it was. You can use a command like the following to move the mouse to the button:

```
$if($1,mouse absmove screen $1)
```